

SPARQL: тройки (triplets)

Евгений Паренченков, 22605

Что такое тройка?

- Тройка (семантический триплет) – это субъект («ребёнок Баха»), предикат («отец субъекта») и объект («Иоганн Себастьян Бах»), описывает взаимоотношения между объектами/понятиями
- В Wikidata Query Service тройки описываются внутри конструкции WHERE
- Шаблон: <subject> <predicate> <object>
- Все элементы тройки могут быть как константами, так и переменными

```
# ?child father Bach
?child wdt:P22 wd:Q1339.
```


Запись значения свойства в переменную

Чтобы получить дополнительную информацию об объекте, достаточно связать его предикатами с переменными

```
SELECT ?child ?childLabel ?genderLabel ?birth_date ?date_of_death
WHERE
{
  ?child wdt:P22 wd:Q1339.# ?child has father Bach
  ?child wdt:P21 ?gender.
  ?child wdt:P569 ?birth_date.
  ?child wdt:P570 ?date_of_death.
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
```

Получаем информацию о тех детях Баха, у которых известны пол, даты рождения и смерти

ОПЦИОНАЛЬНЫЕ СВОЙСТВА

У некоторых объектов свойство может быть не заполненным; можно получить их, обернув тройки в OPTIONAL

```
SELECT DISTINCT ?child ?childLabel ?genderLabel ?birth_date ?date_of_death
WHERE
{
  ?child wdt:P22 wd:Q76.# ?child has father Obama
  OPTIONAL{ ?child wdt:P21 ?gender. }
  OPTIONAL{ ?child wdt:P569 ?birth_date. }
  OPTIONAL{ ?child wdt:P570 ?date_of_death. }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
```

Пол, даты рождения и смерти людей могут быть не известны,
но мы всё равно получим данные о них

«Сложные» тройки

Можно получать сложные взаимоотношения путём добавления предикатов к переменным

```
SELECT ?child ?childLabel ?placeofbirthLabel ?coordinates
WHERE
{
  ?child wdt:P22 wd:Q1339.# ?child has father Bach
  ?child wdt:P19 ?placeofbirth.
  ?placeofbirth wdt:P625 ?coordinates.
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
ORDER BY ?placeofbirthLabel
```

Получаем список детей Баха и их мест рождения, а у мест рождения – координаты

Предикаты как переменные

Можно получать список взаимоотношений между объектами Викиданных, передавая предикат как переменную, а не как идентификатор

```
SELECT ?predicate ?pLabel
WHERE
{
  wd:Q57225 ?predicate wd:Q1339.          # Johann Christoph Friedrich Bach ?predicate Johann Sebastian Bach

  BIND( IRI(REPLACE( STR(?predicate), "prop/direct/", "entity/" )) AS ?p).
  # or ?p wikibase:directClaim ?predicate.
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
```

Получим отношения «отец» и «студент»

Запросы с двумя предикатами

```
SELECT ?predicate ?object
WHERE
{
  wd:Q1339 ?predicate ?object.      # Bach
}
```

Список всех свойств объекта «Иоганн Себастьян Бах» и их значений

Запросы с тремя предикатами

```
SELECT ?subject ?predicate ?object
WHERE
{
  ?subject ?predicate ?object.
  ?subject wdt:P22 wd:Q1339.      # subject has father  Bach
}
ORDER BY ?subject ?predicate ?object
LIMIT 10000
```

Вся информация о всех детях Баха в WD

Осторожно: такие запросы могут выполняться очень долго – лимитируем вывод либо используем их для небольших БД

Предикаты для получения метаданных

```
SELECT ?subject ?subjectLabel ?datemodified ?statements ?sitelinks
WHERE
{
  ?subject wdt:P22 wd:Q1339.          # subject has father Bach
  ?subject schema:dateModified ?datemodified.
  ?subject wikibase:statements ?statements.
  ?subject wikibase:sitelinks ?sitelinks.
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
```

Получаем количество утверждений, даты изменения и количество ссылок для объекта каждого из детей Баха

Спасибо за внимание!

